Solving for c_{in}, $j_{\text{in}}(= D\,\nabla c_{\text{in}})$ inside an active droplet with linearized chemical reactions for any generic reaction flux $\mathrm{s}(c_{\text{in}})$.
Solving for c_{out}, $j_{\text{out}}(= D\,\nabla c_{\text{out}})$ inside the shell for an active/passive droplet with linearized chemical reactions for any generic reaction flux $\mathrm{s}(c_{\text{out}})$.

Note: Cells highlighted in green indicate relevant parts implemented in the code.

Inside the droplet:

We aim to solve the steady state reaction-diffusion equation: $D\,\nabla^2 c_{\text{in}} + \mathrm{s}(c_{\text{in}}) = 0$ for any generic reaction flux scheme $\mathrm{s}(c_{\text{in}})$ inside the droplet. (Note that for the passive case, $c_{\text{in}} = c_{\text{in}}^{\text{eq}}$ and hence $j_{\text{in}} = 0$).

We can linearize the generic reaction flux $\mathrm{s}(c_{\text{in}})$ around c_{in}^{0} as:

$\mathrm{s}(c_{\text{in}}) \approx \mathrm{s}(c_{\text{in}}^{0}) - \mathrm{s}`\,(c_{\text{in}}^{0})\,(c_{\text{in}} - c_{\text{in}}^{0}) \approx \mathrm{s}(c_{\text{in}}^{0}) - k_{\text{in}}\,(c_{\text{in}} - c_{\text{in}}^{0})$

Denoting $\mathrm{s}(c_{\text{in}}^{0})$ as ScZeroIn and k_{in} as k, where k>0 (Refer to Review, *Eqs*. 4.15).

We then solve $D\,\nabla^2 c_{\text{in}} +$ ScZeroIn - k (c_{in} - cZeroIn) = 0 analytically for c_{in} for 1, 2 and 3 dimensions.

We integrate (- D∇c_{in}) at r = R over the surface of the droplet to get the integrated flux J_{in}.

Finally, the mean flux inside the droplet is calculated as J_{in}/(Volume of the droplet), which is used as an input to 'reaction_inside' in the code.

———

Outside the droplet = inside the shell:

We aim to solve the steady state reaction-diffusion equation: $D\,\nabla^2 c_{out} + s(c_{out}) = 0$ for any generic reaction flux scheme $s(c_{out})$ inside each shell sector. (Note that for the passive case, we only solve $D\,\nabla^2 c_{out} = 0$).

Generally, as the shell thickness L is a simulation parameter (and can be large or small as compared to the droplet radius R), c_{out} inside the shell sector can vary a lot spatially. Hence, as a first approximation, we assume the reaction flux $s(c_{out})$ to be a linear function of c_{out}.

We then approximate $s(c_{out})$ as $s(c_{out}) \approx A - k\,c_{out}$, where $k > 0$ (Refer to Review, Eqs. 4.15)

Note that $A - k\,c_{out}$ is used as an input to 'reaction_outside' in the code.

We solve for A and k from the following two equations:

1. $s(c_{out})$ at $r = R$: $s(\,c_{out}^{eq}) = A - k\,c_{out}^{eq}$

2. $s(c_{out})$ at $r = R + L$: $s(cfar) = A - k\,c_{far}$

After determining A and k, we solve $D\,\nabla^2 c_{out} + A - k\,c_{out} = 0$ analytically for c_{out} 1, 2 and 3 dimensions. We then integrate $(-\,D\nabla c_{out})$ at $r = R$ over the surface of the droplet to get the integrated flux J_{out}.

3D: Inside the droplet:

In[ ]:=
```
ClearAll["Global`*"]
```

In[ ]:=
```
$Assumptions = Diff > 0 && k > 0 && cZeroIn > 0 && cEqIn > 0 && R > 0 && ξ > 0;
```

Solve the stationary state equation for 3D droplets

In[ ]:=
```
sol3DInside[r_] =
  FullSimplify[FullSimplify[c[r] /. First@DSolve[{0 == Diff * Laplacian[
          c[r], {r, ϕ, θ}, "Spherical"] + ScZeroIn - k * (c[r] - cZeroIn),
        Derivative[1][c][0] == 0, c[R] == cEqIn}, c, r]] /. Diff → k * ξ^2]
```

Out[ ]=
$$\text{cZeroIn} + \frac{\text{ScZeroIn}}{k} - \frac{R\,(-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn})\,\text{Csch}\left[\frac{R}{\xi}\right]\,\text{Sinh}\left[\frac{r}{\xi}\right]}{k\,r}$$

Check the boundary conditions

In[ ]:=
```
FullSimplify@sol3DInside[R]
```
Out[ ]= cEqIn

In[ ]:=
```
Limit[D[sol3DInside[r], r], r → 0]
```
Out[ ]= 0

Check the solution by plugging it in the Reaction-Diffusion equation

In[ ]:=
```
FullSimplify[
 FullSimplify[Diff * Laplacian[sol3DInside[r], {r, ϕ, θ}, "Spherical"] +
    ScZeroIn - k * (sol3DInside[r] - cZeroIn)] /. Diff → k * ξ^2]
```
Out[ ]= 0

Calculate surface area integrated fluxes at the droplet surface

In[ ]:=
```
FluxInside3D = FullSimplify[
  FullSimplify[-4 π * R^2 * Diff * D[sol3DInside[r], r] /. r → R] /. Diff → k * ξ^2]
```
Out[ ]= $4\,\pi\,R\,(\text{cEqIn}\,k - \text{cZeroIn}\,k - \text{ScZeroIn})\,\xi\left(\xi - R\,\text{Coth}\left[\frac{R}{\xi}\right]\right)$

In[ ]:=
```
FluxInside3DperVolume [R_] = FullSimplify[FluxInside3D / ((4 / 3) π * R * R * R)]
```
Out[ ]= $\frac{3\,(\text{cEqIn}\,k - \text{cZeroIn}\,k - \text{ScZeroIn})\,\xi\left(\xi - R\,\text{Coth}\left[\frac{R}{\xi}\right]\right)}{R^2}$

In[ ]:=
```
Normal[Series[FluxInside3DperVolume[R], {R, 0, 4}]]
```
Out[ ]= $-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn} - \frac{2\,R^4\,(\text{cEqIn}\,k - \text{cZeroIn}\,k - \text{ScZeroIn})}{315\,\xi^4} + \frac{R^2\,(\text{cEqIn}\,k - \text{cZeroIn}\,k - \text{ScZeroIn})}{15\,\xi^2}$

Check if ξ -> ∞ and k -> 0 goes back to FluxInside = ScZeroIn

In[ ]:=
```
temp[F_] =
 FullSimplify[FullSimplify[FluxInside3D / ((4 / 3) π * R * R * R)] /. R → F * ξ]
```
Out[ ]= $-\frac{3\,(\text{cEqIn}\,k - \text{cZeroIn}\,k - \text{ScZeroIn})\,(-1 + F\,\text{Coth}[F])}{F^2}$

In[ ]:=
```
Limit[FullSimplify[Limit[temp[F], F → 0]], k → 0]
```
Limit: Warning: Assumptions that involve the limit variable are ignored.

Out[ ]= ScZeroIn

3D: Outside the droplet = Inside the shell:

```
In[ ]:= ClearAll["Global`*"]

In[ ]:= $Assumptions = k > 0 && cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0

Out[ ]= k > 0 && cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0
```

We solve for A and B as s(c_{out}) at r = R = sOutceqout and s(c_{out}) at r = R + L = sOutcfar

```
In[ ]:= eq1 = A - k * cEqOut == sOutceqout;
        eq2 = A - k * cfar == sOutcfar;
        NSolve[{eq1, eq2}, {A, k}]
```

$$\text{Out}[\;]= \left\{\left\{A \to \frac{1. \times (1.\,\text{cfar sOutceqout} - 1.\,\text{cEqOut sOutcfar})}{-1.\,\text{cEqOut} + 1.\,\text{cfar}},\; k \to -\frac{1. \times (1.\,\text{sOutceqout} - 1.\,\text{sOutcfar})}{1.\,\text{cEqOut} - 1.\,\text{cfar}}\right\}\right\}$$

$$\text{In}[\;]:= A \to \frac{(\text{cfar sOutceqout} - \text{cEqOut sOutcfar})}{-\text{cEqOut} + \text{cfar}}$$

$$\text{Out}[\;]= A \to \frac{\text{cfar sOutceqout} - \text{cEqOut sOutcfar}}{-\text{cEqOut} + \text{cfar}}$$

$$\text{In}[\;]:= k \to -\frac{(\text{sOutceqout} - \text{sOutcfar})}{\text{cEqOut} - \text{cfar}}$$

$$\text{Out}[\;]= k \to -\frac{\text{sOutceqout} - \text{sOutcfar}}{\text{cEqOut} - \text{cfar}}$$

Solve the stationary state equation for 3D droplets

```
In[ ]:= sol3DOutside[r_] = FullSimplify[c[r] /.
          First@DSolve[{0 == Diff * Laplacian[c[r], {r, ϕ, θ}, "Spherical"] + A - k * c[r],
             c[R] == cEqOut, c[R + L] == cfar}, c, r] /. Diff → k * ξ^2]
```

$$\text{Out}[\;]= \frac{1}{k\,r} e^{L/\xi} \left(-1 + \text{Coth}\left[\frac{L}{\xi}\right]\right) \left(A\,r\,\text{Sinh}\left[\frac{L}{\xi}\right] - (A - \text{cfar}\,k)\,(L + R)\,\text{Sinh}\left[\frac{r - R}{\xi}\right] + (-A + \text{cEqOut}\,k)\,R\,\text{Sinh}\left[\frac{L - r + R}{\xi}\right]\right)$$

Check the boundary conditions

```
In[ ]:= FullSimplify@sol3DOutside[R]

Out[ ]= cEqOut

In[ ]:= FullSimplify@sol3DOutside[R + L]

Out[ ]= cfar
```

Check the solution by plugging it in the Reaction-Diffusion equation

```
In[•]:= FullSimplify[Diff * Laplacian[sol3DOutside[r], {r, ϕ, θ}, "Spherical"] +
          A - k * sol3DOutside[r]] /. Diff → k * ξ^2
Out[•]= 0
```

Calculate surface area integrated fluxes at the droplet surface

```
In[•]:= FluxOutside3D =
         FullSimplify[FullSimplify[-4 π * R^2 * Diff * D[sol3DOutside[r], r] /. r → R]]
```

$$\text{Out[•]=}\ \frac{4\,\text{Diff}\,\pi\,R\left(-\left((A-\text{cEqOut}\,k)\left(\xi+R\,\text{Coth}\left[\frac{L}{\xi}\right]\right)\right)+(A-\text{cfar}\,k)\,(L+R)\,\text{Csch}\left[\frac{L}{\xi}\right]\right)}{k\,\xi}$$

3D: Outside the droplet = Inside the shell when ceqout ~ cfar (when k → 0):

```
In[•]:= ClearAll["Global`*"]
In[•]:= $Assumptions = cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0
Out[•]= cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0
```

Solve the stationary state equation for 3D droplets

```
In[•]:= sol3DOutside[r_] = FullSimplify[
          c[r] /. First@DSolve[{0 == Diff * Laplacian[c[r], {r, ϕ, θ}, "Spherical"] + A,
              c[R] == cEqOut, c[R + L] == cfar}, c, r]]
```

$$\text{Out[•]=}\ \frac{6\,\text{cfar}\,\text{Diff}\,(r-R)\,(L+R)+(L-r+R)\,(6\,\text{cEqOut}\,\text{Diff}\,R+A\,L\,(r-R)\,(L+r+2\,R))}{6\,\text{Diff}\,L\,r}$$

Check the boundary conditions

```
In[•]:= FullSimplify@sol3DOutside[R]
Out[•]= cEqOut
In[•]:= FullSimplify@sol3DOutside[R + L]
Out[•]= cfar
```

Check the solution by plugging it in the Reaction-Diffusion equation

```
In[•]:= FullSimplify[Diff * Laplacian[sol3DOutside[r], {r, ϕ, θ}, "Spherical"] + A]
Out[•]= 0
```

Calculate surface area integrated fluxes at the droplet surface

```
In[*]:= FluxOutside3D =
         FullSimplify[FullSimplify[-4 π * R^2 * Diff * D[sol3DOutside[r], r] /. r → R]]
```

Out[*]= $-\frac{2\,\pi\,R\,\left(-6\,cEqOut\,Diff\,(L+R)+6\,cfar\,Diff\,(L+R)+A\,L^{2}\,(L+3\,R)\right)}{3\,L}$

3D: Outside the droplet = Inside the shell (Passive droplet):

```
In[*]:= ClearAll["Global`*"]

In[*]:= $Assumptions = cEqOut > 0 && L > 0 && Diff > 0 && R > 0

Out[*]= cEqOut > 0 && L > 0 && Diff > 0 && R > 0
```

Solve the stationary state equation for 3D droplets

```
In[*]:= sol3DOutsidePassive[r_] = FullSimplify[
          c[r] /. First@DSolve[{0 == Diff * Laplacian[c[r], {r, ϕ, θ}, "Spherical"],
              c[R] == cEqOut, c[R + L] == cfar}, c, r]]
```

Out[*]= $\frac{cfar\,(r-R)\,(L+R)+cEqOut\,R\,(L-r+R)}{L\,r}$

Check the boundary conditions

```
In[*]:= FullSimplify@sol3DOutsidePassive[R]

Out[*]= cEqOut

In[*]:= FullSimplify@sol3DOutsidePassive[R + L]

Out[*]= cfar
```

Check the solution by plugging it in the Diffusion equation

```
In[*]:= FullSimplify[Diff * Laplacian[sol3DOutsidePassive[r], {r, ϕ, θ}, "Spherical"]]

Out[*]= 0
```

Calculate surface area integrated fluxes at the droplet surface

```
In[*]:= FluxOutside3DPassive = FullSimplify[
          FullSimplify[-4 π * R^2 * Diff * D[sol3DOutsidePassive[r], r] /. r → R]]
```

Out[*]= $\frac{4\,(cEqOut-cfar)\,Diff\,\pi\,R\,(L+R)}{L}$

In[ ]:=

In[ ]:=

2D: Inside the droplet:

```
In[ ]:= ClearAll["Global`*"]

In[ ]:= $Assumptions = Diff > 0 && k > 0 && cZeroIn > 0 && cEqIn > 0 && R > 0 && ξ > 0;
```

Solve the stationary state equation for 2D droplets

```
In[ ]:= sol2DInside[r_] = FullSimplify[FullSimplify[c[r] /. First@DSolve[
          {0 == Diff * Laplacian[c[r], {r, ϕ}, "Polar"] + ScZeroIn - k * (c[r] - cZeroIn),
           Derivative[1][c][0] == 0, c[R] == cEqIn}, c, r]] /. Diff → k * (ξ^2)]
```

Out[ ]= $\text{cZeroIn} + \dfrac{\text{ScZeroIn}}{\text{k}} + \dfrac{(\text{cEqIn k} - \text{cZeroIn k} - \text{ScZeroIn})\ \text{BesselI}\left[0, \frac{r}{\xi}\right]}{\text{k BesselI}\left[0, \frac{R}{\xi}\right]}$

Check the boundary conditions

```
In[ ]:= FullSimplify@sol2DInside[R]

Out[ ]= cEqIn

In[ ]:= Limit[D[sol2DInside[r], r], r → 0]

Out[ ]= 0
```

Check the solution by plugging it in the Reaction-Diffusion equation

```
In[ ]:= FullSimplify[
         FullSimplify[Diff * Laplacian[sol2DInside[r], {r, ϕ}, "Polar"] + ScZeroIn -
            k * (sol2DInside[r] - cZeroIn)] /. Diff → k * (ξ^2)]

Out[ ]= 0
```

Calculate surface area integrated fluxes at the droplet surface

In[∘]:=
```
FluxInside2D = FullSimplify[
  FullSimplify[-2 π * R * Diff * D[sol2DInside[r], r] /. r → R] /. Diff → k * (ξ^2)]
```

Out[∘]=
$$\frac{2\,\pi\,R\,(-\text{cEqIn}\,k+\text{cZeroIn}\,k+\text{ScZeroIn})\,\xi\,\text{BesselI}\left[1,\frac{R}{\xi}\right]}{\text{BesselI}\left[0,\frac{R}{\xi}\right]}$$

In[∘]:=
```
FluxInside2DperVolume[R_] = FullSimplify[FluxInside2D / (π * R * R)]
```

Out[∘]=
$$\frac{2\,(-\text{cEqIn}\,k+\text{cZeroIn}\,k+\text{ScZeroIn})\,\xi\,\text{BesselI}\left[1,\frac{R}{\xi}\right]}{R\,\text{BesselI}\left[0,\frac{R}{\xi}\right]}$$

In[∘]:=
```
Normal[Series[FluxInside2DperVolume[R], {R, 0, 4}]]
```

Out[∘]=
$$\left(-\text{cEqIn}\,k+\text{cZeroIn}\,k+\text{ScZeroIn}+\frac{R^4\,(-\text{cEqIn}\,k+\text{cZeroIn}\,k+\text{ScZeroIn})}{48\,\xi^4}-\frac{R^2\,(-\text{cEqIn}\,k+\text{cZeroIn}\,k+\text{ScZeroIn})}{8\,\xi^2}\right)$$

Check if ξ -> ∞ and k -> 0 goes back to FluxInside = ScZeroIn

In[∘]:=
```
temp[F_] = FullSimplify[FullSimplify[FluxInside2D / (π * R * R)] /. R → F * ξ]
```

Out[∘]=
$$\frac{2\,(-\text{cEqIn}\,k+\text{cZeroIn}\,k+\text{ScZeroIn})\,\text{BesselI}[1,F]}{F\,\text{BesselI}[0,F]}$$

In[∘]:=
```
Limit[FullSimplify[Limit[temp[F], F → 0]], k → 0]
```

••• Limit: Warning: Assumptions that involve the limit variable are ignored.

Out[∘]= ScZeroIn

2D: Outside the droplet = Inside the shell:

In[∘]:=
```
ClearAll["Global`*"]
```

In[∘]:=
```
$Assumptions = k > 0 && cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0
```

Out[∘]= k > 0 && cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0

We solve for A and B as s(c_{out}) at r = R = sOutceqout and s(c_{out}) at r = R + L = sOutcfar

In[∘]:=
```
eq1 = A - k * cEqOut == sOutceqout;
eq2 = A - k * cfar == sOutcfar;
NSolve[{eq1, eq2}, {A, k}]
```

Out[∘]=
$$\left\{\left\{A \to \frac{1. \times (1.\ \text{cfar sOutceqout} - 1.\ \text{cEqOut sOutcfar})}{-1.\ \text{cEqOut} + 1.\ \text{cfar}},\ k \to -\frac{1. \times (1.\ \text{sOutceqout} - 1.\ \text{sOutcfar})}{1.\ \text{cEqOut} - 1.\ \text{cfar}}\right\}\right\}$$

In[∘]:=
$$A \to \frac{(\text{cfar sOutceqout} - \text{cEqOut sOutcfar})}{-\text{cEqOut} + \text{cfar}}$$

Out[∘]=
$$A \to \frac{\text{cfar sOutceqout} - \text{cEqOut sOutcfar}}{-\text{cEqOut} + \text{cfar}}$$

In[∘]:=
$$k \to -\frac{(\text{sOutceqout} - \text{sOutcfar})}{\text{cEqOut} - \text{cfar}}$$

Out[∘]=
$$k \to -\frac{\text{sOutceqout} - \text{sOutcfar}}{\text{cEqOut} - \text{cfar}}$$

Solve the stationary state equation for 2D droplets

In[∘]:=
```
solI2DOutside[r_] = FullSimplify[
  c[r] /. First@DSolve[{0 == Diff * Laplacian[c[r], {r, ϕ}, "Polar"] + A - k * c[r],
     c[R] == cEqOut, c[R + L] == cfar}, c, r] /. Diff → k * ξ^2]
```

Out[∘]=
$$\left(\pi \left(\text{BesselI}\left[0, \frac{L+R}{\xi}\right] \left((-A + \text{cEqOut}\ k)\ \text{BesselY}\left[0, -\frac{i\, r}{\xi}\right] + A\ \text{BesselY}\left[0, -\frac{i\, R}{\xi}\right]\right) + \text{BesselI}\left[0, \frac{R}{\xi}\right] \left((A - \text{cfar}\ k)\ \text{BesselY}\left[0, -\frac{i\, r}{\xi}\right] - A\ \text{BesselY}\left[0, -\frac{i\,(L+R)}{\xi}\right]\right) + \text{BesselI}\left[0, \frac{r}{\xi}\right] \left((-A + \text{cfar}\ k)\ \text{BesselY}\left[0, -\frac{i\, R}{\xi}\right] + (A - \text{cEqOut}\ k)\ \text{BesselY}\left[0, -\frac{i\,(L+R)}{\xi}\right]\right)\right)\right) \Big/ \left(2 \left(-k\ \text{BesselI}\left[0, \frac{L+R}{\xi}\right] \text{BesselK}\left[0, \frac{R}{\xi}\right] + k\ \text{BesselI}\left[0, \frac{R}{\xi}\right] \text{BesselK}\left[0, \frac{L+R}{\xi}\right]\right)\right)$$

In[•]:=
```
sol2DOutside[r_] =
 FullSimplify[solI2DOutside[r] /. BesselY[0, -I * x_] :> -2 / π * BesselK[0, x]]
```

Out[•]=
$$\left(\text{BesselI}\left[0, \frac{L+R}{\xi}\right]\left((-A + \text{cEqOut } k)\ \text{BesselK}\left[0, \frac{r}{\xi}\right] + A\ \text{BesselK}\left[0, \frac{R}{\xi}\right]\right) + \text{BesselI}\left[0, \frac{R}{\xi}\right]\left((A - \text{cfar } k)\ \text{BesselK}\left[0, \frac{r}{\xi}\right] - A\ \text{BesselK}\left[0, \frac{L+R}{\xi}\right]\right) + \text{BesselI}\left[0, \frac{r}{\xi}\right]\left((-A + \text{cfar } k)\ \text{BesselK}\left[0, \frac{R}{\xi}\right] + (A - \text{cEqOut } k)\ \text{BesselK}\left[0, \frac{L+R}{\xi}\right]\right)\right) \Big/ \left(k\ \text{BesselI}\left[0, \frac{L+R}{\xi}\right]\text{BesselK}\left[0, \frac{R}{\xi}\right] - k\ \text{BesselI}\left[0, \frac{R}{\xi}\right]\text{BesselK}\left[0, \frac{L+R}{\xi}\right]\right)$$

Check the boundary conditions

In[•]:=
```
FullSimplify[sol2DOutside[R]]
```

Out[•]=
```
cEqOut
```

In[•]:=
```
FullSimplify[sol2DOutside[R + L]]
```

Out[•]=
```
cfar
```

Check the solution by plugging it in the Reaction-Diffusion equation

In[•]:=
```
FullSimplify[Diff * Laplacian[sol2DOutside[r], {r, ϕ}, "Polar"] +
   A - k * sol2DOutside[r]] /. Diff → k * ξ^2
```

Out[•]=
```
0
```

Calculate surface area integrated fluxes at the droplet surface

In[•]:=
```
FluxOutside2D =
 FullSimplify[FullSimplify[-2 π * R * Diff * D[sol2DOutside[r], r] /. r → R]]
```

Out[•]=
$$\left(2\ \text{Diff}\ \pi\left((A - \text{cfar } k)\ \xi + (-A + \text{cEqOut } k)\ R\left(\text{BesselI}\left[1, \frac{R}{\xi}\right]\text{BesselK}\left[0, \frac{L+R}{\xi}\right] + \text{BesselI}\left[0, \frac{L+R}{\xi}\right]\text{BesselK}\left[1, \frac{R}{\xi}\right]\right)\right)\right) \Big/ \left(k\ \xi\left(\text{BesselI}\left[0, \frac{L+R}{\xi}\right]\text{BesselK}\left[0, \frac{R}{\xi}\right] - \text{BesselI}\left[0, \frac{R}{\xi}\right]\text{BesselK}\left[0, \frac{L+R}{\xi}\right]\right)\right)$$

2D: Outside the droplet = Inside the shell when ceqout ~ cfar (when k → 0):

In[•]:=
```
ClearAll["Global`*"]
```

In[•]:=
```
$Assumptions = cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0
```

Out[•]=
```
cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0
```

Solve the stationary state equation for 2D droplets

In[∘]:=
```
sol2DOutside[r_] =
  FullSimplify[c[r] /. First@DSolve[{0 == Diff * Laplacian[c[r], {r, ϕ}, "Polar"] + A,
      c[R] == cEqOut, c[R + L] == cfar}, c, r]]
```

Out[∘]=
$$\frac{1}{4\,\text{Diff}\,\text{Log}\left[\frac{R}{L+R}\right]}\big((4\,\text{cEqOut}\,\text{Diff}-4\,\text{cfar}\,\text{Diff}-A\,L\,(L+2\,R))\,\text{Log}[r]+(4\,\text{cfar}\,\text{Diff}+A\,(L-r+R)\,(L+r+R))\,\text{Log}[R]+(-4\,\text{cEqOut}\,\text{Diff}+A\,(r-R)\,(r+R))\,\text{Log}[L+R]\big)$$

Check the boundary conditions

In[∘]:=
```
FullSimplify@sol2DOutside[R]
```

Out[∘]= cEqOut

In[∘]:=
```
FullSimplify@sol2DOutside[R + L]
```

Out[∘]= cfar

Check the solution by plugging it in the Reaction-Diffusion equation

In[∘]:=
```
FullSimplify[Diff * Laplacian[sol2DOutside[r], {r, ϕ}, "Polar"] + A]
```

Out[∘]= 0

Calculate surface area integrated fluxes at the droplet surface

In[∘]:=
```
FluxOutside2D =
  FullSimplify[FullSimplify[-2 π * R * Diff * D[sol2DOutside[r], r] /. r → R]]
```

Out[∘]=
$$A\,\pi\,R^2+\frac{\pi\,(-4\,\text{cEqOut}\,\text{Diff}+4\,\text{cfar}\,\text{Diff}+A\,L\,(L+2\,R))}{2\,\text{Log}\left[\frac{R}{L+R}\right]}$$

2D: Outside the droplet = Inside the shell (Passive droplet):

In[∘]:=
```
ClearAll["Global`*"]
```

In[∘]:=
```
$Assumptions = cEqOut > 0 && L > 0 && Diff > 0 && R > 0
```

Out[∘]= cEqOut > 0 && L > 0 && Diff > 0 && R > 0

Solve the stationary state equation for 2D droplets

In[∘]:=
```
sol2DOutsidePassive[r_] =
  FullSimplify[c[r] /. First@DSolve[{0 == Diff * Laplacian[c[r], {r, ϕ}, "Polar"],
      c[R] == cEqOut, c[R + L] == cfar}, c, r]]
```

Out[∘]=
$$\frac{(\text{cEqOut}-\text{cfar})\,\text{Log}[r]+\text{cfar}\,\text{Log}[R]-\text{cEqOut}\,\text{Log}[L+R]}{\text{Log}\left[\frac{R}{L+R}\right]}$$

Check the boundary conditions

In[ ]:=
```
FullSimplify@sol2DOutsidePassive[R]
```
Out[ ]=
```
cEqOut
```

In[ ]:=
```
FullSimplify@sol2DOutsidePassive[R + L]
```
Out[ ]=
```
cfar
```

Check the solution by plugging it in the Diffusion equation

In[ ]:=
```
FullSimplify[Diff * Laplacian[sol2DOutsidePassive[r], {r, ϕ}, "Polar"]]
```
Out[ ]=
```
0
```

Calculate surface area integrated fluxes at the droplet surface

In[ ]:=
```
FluxOutside2DPassive =
  FullSimplify[FullSimplify[-2 π * R * Diff * D[sol2DOutsidePassive[r], r] /. r → R]]
```
Out[ ]=

$$\frac{2\,(-\text{cEqOut} + \text{cfar})\,\text{Diff}\,\pi}{\text{Log}\left[\frac{R}{L+R}\right]}$$

In[ ]:=

1D: Inside the droplet:

In[ ]:=
```
ClearAll["Global`*"]
```

In[ ]:=
```
$Assumptions = Diff > 0 && k > 0 && cZeroIn > 0 && cEqIn > 0 && R > 0 && ξ > 0;
```

Solve the stationary state equation for 1D droplets

In[ ]:=
```
sol1DInside[x_] = FullSimplify[c[x] /. First@DSolve[
      {0 == Diff * Laplacian[c[x], {x}, "Cartesian"] + ScZeroIn - k * (c[x] - cZeroIn),
       Derivative[1][c][0] == 0, c[R] == cEqIn}, c, x] /. Diff → k * (ξ^2)]
```
Out[ ]=

$$\frac{\text{cZeroIn}\,k + \text{ScZeroIn} - (-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn})\,\text{Cosh}\left[\frac{x}{\xi}\right]\,\text{Sech}\left[\frac{R}{\xi}\right]}{k}$$

Check the boundary conditions

In[*]:=
```
FullSimplify@sol1DInside[R]
```
Out[*]=
```
cEqIn
```
In[*]:=
```
Limit[D[sol1DInside[x], x], x → 0]
```
Out[*]=
```
0
```

Check the solution by plugging it in the Reaction-Diffusion equation

In[*]:=
```
FullSimplify[
 FullSimplify[Diff * Laplacian[sol1DInside[x], {x}, "Cartesian"] + ScZeroIn -
   k * (sol1DInside[x] - cZeroIn)] /. Diff → k * (ξ^2)]
```
Out[*]=
```
0
```

Calculate surface area integrated fluxes at the droplet surface

In[*]:=
```
FluxInside1D = FullSimplify[
  FullSimplify[-2 * Diff * D[sol1DInside[x], x] /. x → R] /. Diff → k * (ξ^2)]
```
Out[*]=
$$2\,(-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn})\,\xi\,\text{Tanh}\left[\frac{R}{\xi}\right]$$

In[*]:=
```
FluxInside1DperVolume[R_] = FullSimplify[FluxInside1D / (2 * R)]
```
Out[*]=
$$\frac{(-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn})\,\xi\,\text{Tanh}\left[\frac{R}{\xi}\right]}{R}$$

In[*]:=
```
Normal[Series[FluxInside1DperVolume[R], {R, 0, 4}]]
```
Out[*]=
$$\left(-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn} + \frac{2\,R^4\,(-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn})}{15\,\xi^4} - \frac{R^2\,(-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn})}{3\,\xi^2}\right)$$

Check if ξ -> ∞ and k -> 0 goes back to FluxInside = ScZeroIn

In[*]:=
```
temp[F_] = FullSimplify[FullSimplify[FluxInside1D / (2 * R)] /. R → F * ξ]
```
Out[*]=
$$\frac{(-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn})\,\text{Tanh}[F]}{F}$$

In[*]:=
```
Limit[FullSimplify[Limit[temp[F], F → 0]], k → 0]
```
··· Limit: Warning: Assumptions that involve the limit variable are ignored.

Out[*]=
```
ScZeroIn
```

1D: Outside the droplet = Inside the shell:

In[∘]:=
```
ClearAll["Global`*"]
```

In[∘]:=
```
$Assumptions = k > 0 && cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0
```

Out[∘]= $k > 0 \,\&\&\, cEqOut > 0 \,\&\&\, L > 0 \,\&\&\, \xi > 0 \,\&\&\, Diff > 0 \,\&\&\, R > 0$

We solve for A and B as s(c_{out}) at r = R = sOutceqout and s(c_{out}) at r = R + L = sOutcfar

In[∘]:=
```
eq1 = A - k * cEqOut == sOutceqout;
eq2 = A - k * cfar == sOutcfar;
NSolve[{eq1, eq2}, {A, k}]
```

Out[∘]= $$\left\{\left\{A \to \frac{1. \times (1.\, cfar\, sOutceqout - 1.\, cEqOut\, sOutcfar)}{-1.\, cEqOut + 1.\, cfar},\; k \to -\frac{1. \times (1.\, sOutceqout - 1.\, sOutcfar)}{1.\, cEqOut - 1.\, cfar}\right\}\right\}$$

In[∘]:= $$A \to \frac{(cfar\, sOutceqout - cEqOut\, sOutcfar)}{-cEqOut + cfar}$$

Out[∘]= $$A \to \frac{cfar\, sOutceqout - cEqOut\, sOutcfar}{-cEqOut + cfar}$$

In[∘]:= $$k \to -\frac{(sOutceqout - sOutcfar)}{cEqOut - cfar}$$

Out[∘]= $$k \to -\frac{sOutceqout - sOutcfar}{cEqOut - cfar}$$

Solve the stationary state equation for 1D droplets

In[∘]:=
```
sol1DOutside[x_] = FullSimplify[
  c[x] /. First@DSolve[{0 == Diff * Laplacian[c[x], {x}, "Cartesian"] + A - k * c[x],
      c[R] == cEqOut, c[R + L] == cfar}, c, x] /. Diff → k * ξ^2]
```

Out[∘]= $$\frac{e^{L/\xi}\left(-1 + \mathrm{Coth}\left[\frac{L}{\xi}\right]\right)\left(A\, \mathrm{Sinh}\left[\frac{L}{\xi}\right] + (A - cfar\, k)\, \mathrm{Sinh}\left[\frac{R-x}{\xi}\right] + (-A + cEqOut\, k)\, \mathrm{Sinh}\left[\frac{L+R-x}{\xi}\right]\right)}{k}$$

Check the boundary conditions

In[∘]:=
```
FullSimplify@sol1DOutside[R]
```

Out[∘]= cEqOut

In[∘]:=
```
FullSimplify@sol1DOutside[R + L]
```

Out[∘]= cfar

Check the solution by plugging it in the Reaction-Diffusion equation

```
In[•]:= FullSimplify[FullSimplify[Diff * Laplacian[sol1DOutside[x], {x}, "Cartesian"] +
        A - k * sol1DOutside[x]] /. Diff → k * ξ^2]
Out[•]= 0
```

Calculate surface area integrated fluxes at the droplet surface

```
In[•]:= FluxOutside1D = FullSimplify[-2 * Diff * D[sol1DOutside[x], x] /. x → R]
```

$$\text{Out[•]=} \quad -\frac{2\,\text{Diff}\left(-A+\text{cfar}\,k+(A-\text{cEqOut}\,k)\,\text{Cosh}\left[\frac{L}{\xi}\right]\right)\text{Csch}\left[\frac{L}{\xi}\right]}{k\,\xi}$$

1D: Outside the droplet = Inside the shell when ceqout ~ cfar (when k → 0):

```
In[•]:= ClearAll["Global`*"]
In[•]:= $Assumptions = cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0
Out[•]= cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0
```

Solve the stationary state equation for 1D droplets

```
In[•]:= sol1DOutside[x_] = FullSimplify[
          c[x] /. First@DSolve[{0 == Diff * Laplacian[c[x], {x}, "Cartesian"] + A,
              c[R] == cEqOut, c[R + L] == cfar}, c, x]]
```

$$\text{Out[•]=} \quad \frac{-((2\,\text{cfar}\,\text{Diff}+A\,L\,(L+R-x))\,(R-x))+2\,\text{cEqOut}\,\text{Diff}\,(L+R-x)}{2\,\text{Diff}\,L}$$

Check the boundary conditions

```
In[•]:= FullSimplify@sol1DOutside[R]
Out[•]= cEqOut
In[•]:= FullSimplify@sol1DOutside[R + L]
Out[•]= cfar
```

Check the solution by plugging it in the Reaction-Diffusion equation

```
In[•]:= FullSimplify[Diff * Laplacian[sol1DOutside[x], {x}, "Cartesian"] + A]
Out[•]= 0
```

Calculate surface area integrated fluxes at the droplet surface

In[•]:=
```
FluxOutside1D =
 FullSimplify[FullSimplify[-2 * Diff * D[sol1DOutside[x], x] /. x → R]]
```

Out[•]=
$$\frac{2\ (\text{cEqOut} - \text{cfar})\ \text{Diff}}{\text{L}} - \text{A L}$$

1D: Outside the droplet = Inside the shell (Passive droplet):

In[•]:=
```
ClearAll["Global`*"]
```

In[•]:=
```
$Assumptions = cEqOut > 0 && L > 0 && l > 0 && Diff > 0 && R > 0
```

Out[•]= cEqOut > 0 && L > 0 && l > 0 && Diff > 0 && R > 0

Solve the stationary state equation for 1D droplets

In[•]:=
```
sol1DOutsidePassive[x_] =
 FullSimplify[c[x] /. First@DSolve[{0 == Diff * Laplacian[c[x], {x}, "Cartesian"],
      c[R] == cEqOut, c[R + L] == cfar}, c, x]]
```

Out[•]=
$$\frac{\text{cEqOut}\ (\text{L} + \text{R} - \text{x}) + \text{cfar}\ (-\text{R} + \text{x})}{\text{L}}$$

Check the boundary conditions

In[•]:=
```
FullSimplify@sol1DOutsidePassive[R]
```

Out[•]= cEqOut

In[•]:=
```
FullSimplify@sol1DOutsidePassive[R + L]
```

Out[•]= cfar

Check the solution by plugging it in the Diffusion equation

In[•]:=
```
FullSimplify[Diff * Laplacian[sol1DOutsidePassive[x], {x}, "Cartesian"]]
```

Out[•]= 0

Calculate surface area integrated fluxes at the droplet surface

In[•]:=
```
FluxOutside1DPassive =
 FullSimplify[FullSimplify[-2 * Diff * D[sol1DOutsidePassive[x], x] /. x → R]]
```

Out[•]=
$$\frac{2\ (\text{cEqOut} - \text{cfar})\ \text{Diff}}{\text{L}}$$